
Introduction
------------------------------------------------------------------------------------------------------------------------------------------------------------------

In this Technical Note you’ll learn how to take advantage of the Developer Hooks
available in the 4D Ajax Framework. Right out of the box, 4D Ajax Framework can
be integrated into most 4D databases to make them Web 2.0 enabled. The basic or
default functionality, however, may not be enough in many 4D Web projects.
Developers may require overrides that allow them to alter existing behavior. This
capability is provided via Developer Hooks in 4D Ajax Framework.

The Technical Note core material covers four main areas of the Developer Hooks:
Developer Created Selections, Developer Defined Window, Callbacks, and Choice
Lists.

Overview
------------------------------------------------------------------------------------------------------------------------------------------------------------------

The 4D Ajax Framework (4DAF) allows 4D developers to add web-browsing
integration into their applications. Using Ajax technology, 4D developers can offer
the end-user a rich Web 2.0 interface that can browse, add, modify, and delete
records from the underlying database. 4D Ajax Framework includes a component
and a full JavaScript framework that is available for integration into existing HTML
pages. Experienced 4D web developers will be able to use the 4DAF API (JavaScript
framework) to quickly add 4D Ajax Framework objects into their existing
HTML/JavaScript pages.

The 4D Ajax Framework is composed of 2 parts: database framework
component and front-end framework files.

Database Framework Component

The database framework component contains the project methods that handle all
4D Ajax Framework web requests and responses.

In 4D 2004 the database framework component consists of 3 types of methods:
Public, Protected and Private.

Public Methods

Public methods begin with the prefix “DAX_DevHook_” and are editable by the
developer. These methods allow the developer to write code that adds to or
overrides the default behavior of the 4D Ajax Framework. The hooks include:
installing Callbacks and their respective event responses, creating and handling
DCS Views, installing and assigning DDWs, Choice List installation and content



overrides, Users and Groups manipulation, Login and Session control, Query
control, Record controls, and setting 4DAF Preferences.

Protected Methods

Protected methods begin with the prefix “DAX_Dev_” and are provided as a set
of utility methods that are callable by the developer. In most cases, the
protected methods will be used inside Public “DAX_DevHook_” methods. The
initialize and shutdown methods are used only within the Database Methods.
These protected methods are non-modifiable.

Private Methods

Private methods are used by 4D Ajax Framework internally and are hidden from
the developer.

Front-end Framework Files

The front-end 4D Ajax Framework files provide a JavaScript library, multiple CSS
themes, and 4D Ajax Framework Client (HTML). These items are necessary for the
Web 2.0 features available in the Web Interface.

Developer Hooks
------------------------------------------------------------------------------------------------------------------------------------------------------------------

The goal of 4D Ajax Framework is to give 4D developers a way to develop a Web
2.0 solution quickly and efficiently. One way to do that is to give them as many
ways as possible to place and implement their own requirements and designs into
the system. The Developer Hooks in 4D Ajax Framework were designed to meet
this need.

Developer Hooks are provided in the form of public component methods. All of the
methods have the prefix “DAX_DevHook_” and are editable by the developer. The
following is a breakdown of the DevHook methods:



Callback installation and event response
Method Names Description

DAX_DevHook_CB_EventFired [Callbacks] This method is called
anytime a field receives an enabled
event. Supported events are On Load
or On Data Change (applicable only
to input fields).

DAX_DevHook_CB_Install [Callbacks] This method is called at
startup to allow the developer to
enable events for specific fields
(applicable only to input fields).

DAX_DevHook_InstallCallBack [Callbacks – Backward compatibility
for pre 1.2 version] This method was
the previous way of adding a Callback
to handle an event. It should no
longer be used, but is kept for
compatibility with version 1.1
(applicable only to input fields).

DCS creation and record handling
Method Names Description

DAX_DevHook_DCS_RecordDelete [DCS] This method is called
whenever 4DAF is asked to delete a
DCS record. It is up to the developer
to take appropriate action.

DAX_DevHook_DCS_RecordSave [DCS] This method is called
whenever 4DAF is asked to save a
DCS record. It is up to the developer
to take appropriate action.

DAX_DevHook_DCS_SetSelection [DCS] This method is called
whenever 4DAF needs the selection
for a DCS View added through the
DAX_DevHook_DCS_ViewAdd
method. The developer populates
their arrays and returns them.

DAX_DevHook_DCS_ViewAdd [DCS] This method is provided for the
Developer to add custom Views to the
4DAF web interface using arrays as
the data source.

DDW installation and assignment
Method Names Description

DAX_DevHook_DDW_Install [DDW] This method is provided for
the Developer to add DDW Views to
the 4DAF web interface.



Choice List installations, assignments and manipulations
Method Names Description

DAX_DevHook_InstallChoiceList [Choice List] This method allows
developers to install or overwrite a
choice list that will be used for a field
(applicable only to input fields).

DAX_DevHook_ListContents [Choice List] This method is provided
for the Developer to override a list
before it is sent to the front-end.

Users and Groups manipulation
Method Names Description

DAX_DevHook_GetGroupsList [Users/Groups] This method is
provided for the Developer to
override the default 4DAF Groups
system. If you have a users/groups
system in place you can override the
4DAF group system here. This should
be used in tandem with
DAX_DevHook_UserInGroup.

DAX_DevHook_UserInGroup [Users/Groups] This method is
provided for the Developer to
override the default 4DAF Groups
system. If you have a users/groups
system in place you can override the
4DAF group system here. This should
be used in tandem with
DAX_DevHook_GetGroupsList.

Login and Session control
Method Names Description

DAX_DevHook_Login [Login] This method is provided for
the Developer to override the default
4DAF login system. The default is to
use the built-in Users and Groups
system.

DAX_DevHook_SessionValidate [Session] This method is provided for
the Developer to override the default
4DAF session management system.
The developer must also override
DAX_DevHook_Login if they are
taking control of session
management.



Query control
Method Names Description

DAX_DevHook_OnQuery [Query] This method is called
whenever 4DAF is about to perform a
query. Developers can perform their
own query in this method. This may
be useful when performing queries on
a separate lookup table or if you wish
to implement 'fuzzy matching' or
Hash based queries.

DAX_DevHook_QueryAdd [Query] This method is provided for
the Developer to add custom queries
to the 4DAF web interface. All custom
queries will be added to the standard
list of queries available on the front-
end admin area. They appear to the
user as tabs in the selection window.

DAX_DevHook_QueryFilter [Query] This method is called right
before the XML that is sent to the
front-end is constructed. If you need
to remove any records from the
selection that was created you can do
so here.

Record control
Method Names Description

DAX_DevHook_DeleteRecord [Record Control] This method is
called whenever 4DAF is about to
delete a record. The developer can
accept or reject the action and also
handle any other actions, such as
logging, that is required.

DAX_DevHook_SaveRecord [Record Control] This method is
called whenever 4DAF is about to
save a record. The developer can
accept or reject the action and also
handle any other actions, such as
logging or deleting related records.

Global Preferences
Method Names Description

DAX_DevHook_Preferences [Global Preferences] This method is
provided for the Developer to
override the default preferences of
the 4DAF system

As mentioned in the introduction, this Technical Note will focus on four areas of the
developer hooks: Developer Created Selections, Developer Defined Windows,
Callbacks, and Choice Lists.



Developer Created Selections
------------------------------------------------------------------------------------------------------------------------------------------------------------------

Developer Created Selections (DCS) allow the developer to create a custom table
(View) in the 4DAF structure. The DCS View appears to the Web front-end the same
way as any 4D table. The goal of DCS is to de-couple the data between the local
database and local structure. It gives the developer freedom in creating a data
selection in any way and from any data source they want.

Unlike physical tables or Views, a DCS structure and selection must be created
programmatically. The developer must compose the DCS structure using arrays.
The content of the arrays will be read into the 4D Ajax Framework structure during
the startup of the application. There are total of 6 arrays required in a DCS View.

ARRAY #1 (Text Array) contains the name of all fields
ARRAY #2 (Longint Array) contains the type of all fields
ARRAY #3 (Boolean Array) contains the unique property of all fields
ARRAY #4 (Boolean Array) contains the mandatory property of all fields
ARRAY #5 (Boolean Array) contains the non-enterable property of all fields
ARRAY #6 (Boolean Array) contains the non-modifiable property of all fields

Given the following 4D table, we can duplicate the table with DCS by composing the
arrays as follows:

A1 {1}:=”FirstName", {2}:=”LastName”, {3}:=” DOB”
A2 {1}:=”Is String Var”, {2}:=”Is String Var”, {3}:=”Is Date”
A3 {1}:=False, {2}:=False, {3}:=False
A4 {1}:=False, {2}:=False, {3}:=False
A5 {1}:=False, {2}:=False, {3}:=False
A6 {1}:=False, {2}:=False, {3}:=False

Note: A DCS table with a specific name will be created only once. Each DCS must have a
unique name.

Create a DCS Table

Based on the above example, I am going to show you how to create the DCS
structure. There are two Developer Hook methods that are involved in the DCS
creation process: DAX_DevHook_DCS_ViewAdd and
DAX_DevHook_DCS_SetSelection. To create the Contacts table as a DCS, we
must first edit the method DAX_DevHook_DCS_ViewAdd:

  ` Method: DAX_DevHook_DCS_ViewAdd
ARRAY TEXT($A1;3)
ARRAY LONGINT($A2;3)
ARRAY BOOLEAN($A3;3) ` False by default
ARRAY BOOLEAN($A4;3) ` False by default
ARRAY BOOLEAN($A5;3) ` False by default



ARRAY BOOLEAN($A6;3) ` False by default

  ` set the names of the columns
$A1{1}:="FirstName"
$A1{2}:="LastName"
$A1{3}:="DOB"

  ` set the data types of the columns
$A2{1}:=Is String Var
$A2{2}:=Is String Var
$A2{3}:=Is Date

  ` Create DCS View named myContacts
$added_b:=DAX_Dev_DCS_AddCustomView ("myContacts";->$A1;->$A2;->$A3;->$A4;-
>$A5;->$A6)

Note: Remember that each statement in 4D must always be on one line. If you see a line
break in the example code (as in the last line above) it is cosmetic only.

The method DAX_Dev_DCS_AddCustomView is a protected component method
that is used to add the composed DCS View into the 4D Ajax Framework structure.

Syntax: DAX_Dev_DCS_AddCustomView                                              
$1 Text DCS Name (Must be a unique name)
$2 Pointer Pointer to a Text array for field names
$3 Pointer Pointer to a Longint array for the field types
$4 Pointer Pointer to a Boolean array for the unique property
$5 Pointer Pointer to a Boolean array for the mandatory property
$6 Pointer Pointer to a Boolean array for the non-enterable property
$7 Pointer Pointer to a Boolean array for the non-modifiable property

At this point the DCS table does not have any data in it yet. It is the developer’s
responsibility to supply it with a set of data. To do this, the developer must
compose a data set inside the method DAX_DevHook_DCS_SetSelection. Here is an
example:

  ` Method: DAX_DevHook_DCS_SetSelection
Case of
  :($1="myContacts")

ARRAY LONGINT(recIDs_al;0)
ARRAY TEXT(fNames_at;0)
ARRAY TEXT(lNames_at;0)
ARRAY DATE(dobs_ad;0)

  ` Populate data into arrays
ALL RECORDS([Contacts])
SELECTION TO ARRAY([Contacts];recIDs _al;[Contacts]FirstName;fNames_at;

[Contacts]LastName;lNames_at;[Contacts]DOB;dobs_ad)

  `Set the selection in DAX in the order defined
  `when we created the View
DAX_Dev_DCS_SetSelection(->recIDs_al;->fNames_at;->lNames_at;->dobs_ad)

End Case

In this method, we are not only populating the data into the fNames_at, lNames_at
and dobs_ad arrays, we also need to generate the record id for each row. The



record id will be used to identify a specific record during record modification and
deletion. When a query request is made for any DCS record this method will be
executed first to setup the initial selection. This selection will then be queried on
with the criteria posted by the Web request.

This method has one parameter. This parameter contains the name of the
requested DCS. In the above example, the Case-of is set up to check if the name of
the requested DCS is “myContacts.” If so, it will generate the initial selection for the
DCS. The same approach will be used for record saving and deleting of the DCS as
well.

Note: Only process or interprocess arrays can be used to create a selection in method
DAX_DevHook_DCS_SetSelection

The last call in the method is DAX_Dev_DCS_SetSelection. This method is
another protected component method that should be used specifically in
DAX_DevHook_DCS_SetSelection method.

Syntax: DAX_Dev_DCS_SetSelection                                                     
$1 Pointer to a Longint array containing unique record IDs (Required)
${2 to N} Pointer to array of DCS field data (where N=number of fields) in

DAX_DevHook_DCS_ViewAdd

Handle record saving and deleting

Each DCS has a specific name associated to it. It will be used to determine which
DCS View the record belongs to. Since the developer defines the DCS name, the
responsibility falls to the developer to trap the name appropriately when a DCS
record is saved or deleted. To handle record saving the developer needs to place
the handler code into the method named DAX_DevHook_DCS_RecordSave. Here
is a simple example of how to handle record saving in the DCS named
“myContacts.”

 ` Method: DAX_DevHook_DCS_RecordSave
$viewName_t:=$1
$recordID_l:=$2
$addRecFieldName_p:=$3
$addRecFieldValue_p:=$4

Case of
  :($viewName_t="myContacts")
    If ($recordID_l=New record )
      CREATE RECORD([Contacts])
    Else
      READ WRITE([Contacts])
      GOTO RECORD([Contacts];$recordID_l)
    End if

    $find_l:=Find in array($addRecFieldName_p->;"FirstName")
    If ($find_l#-1)
      [Contacts]FirstName:=$addRecFieldValue_p->{$find_l}
    End if



    $find_l:=Find in array($addRecFieldName_p->;"LastName")
    If ($find_l#-1)
      [Contacts]LastName:=$addRecFieldValue_p->{$find_l}
    End if

    $find_l:=Find in array($addRecFieldName_p->;"DOB")
    If ($find_l#-1)
      [Contacts]DOB:=4DAFDate_2_4DDate($addRecFieldValue_p->{$find_l})
    End if

    SAVE RECORD([Contacts])

    $recordID_l:=Record number([Contacts])
End case

$0:=$recordID_l

When a date value is sent by the front-end, the value comes in as Text (YYYY-MM-
DD). For DCS, the developer will need to convert the date value into an actual date
value that 4D understands. The following method is a generic method that can be
used in any DCS:

  ` Generic Method: 4DAFDate_2_4DDate
  `    Convert Date value posted through 4DAF front-end into
  `    4D date. The method can be used in any DCS table
  `    with a Date field (column)
  ` Conversion: “YYYY-MM-DD” to MM/DD/YYYY
C_DATE($0)
C_TEXT($1;$date_t;$year_t;$month_t;$day_t)
C_LONGINT($pos_l)
$year_t:="0000"
$month_t:="00"
$day_t:="00"
If (Count parameters>=1)
  $date_t:=$1
  $pos_l:=Position("-";$date_t)
  If ($pos_l>0)
    $year_t:=Substring($date_t;1;$pos_l-1)
    $date_t:=Delete string($date_t;1;$pos_l)
    $pos_l:=Position("-";$date_t)
    If ($pos_l>0)
      $month_t:=Substring($date_t;1;$pos_l-1)
      $day_t:=Delete string($date_t;1;$pos_l)
    End if
  End if
End if
$0:=Date($month_t+"/"+$day_t+"/"+$year_t)

Deleting a DCS record can be done in a similar fashion as saving a record.
DAX_DevHook_DCS_RecordDelete takes two parameters: DCS name and record
id. They will be used to determine the exact record that the user wants to delete.

  ` Method: DAX_DevHook_DCS_RecordDelete
$viewName_t:=$1
$recordID_l:=$2
$deleteRecordAccepted_b:=True
Case of
  :($viewName_t="myContacts")
    If ($recordID_l>No current record )
      READ WRITE([Contacts])



      GOTO RECORD([Contacts];$recordID_l)
    End if
    If (Records in selection([Contacts])=1)
      DELETE RECORD([Contacts])
    Else
      $deleteRecordAccepted_b:=False
      UNLOAD RECORD([Contacts])
    End if
End case

Developer Defined Windows (DDW)
------------------------------------------------------------------------------------------------------------------------------------------------------------------

DDWs give developers the ability to add custom content to the front-end. The
content can be HTML, a URL, or JavaScript code. They can be used to create a
mash-up from multiple sources of Web material, execute a JavaScript call, create a
dynamic HTML blob, or create a dynamic URL based on data from the front-end.

Understanding different types of DDW

A DDW is handled in a similar fashion as any other request that comes into 4D.
When a request comes in, there must be 4D code to handle the request. For a
DDW, however, there are two types of responses: dynamic and static.

A Dynamic DDW is for handling a method that returns a URL or a JavaScript call
from the back-end. Use of a DDW can be varied depending on the functionality.
Here are some examples:

When $0 is a URL

$0:=”http://www.4d.com”

$0:=”http://www.domain.com/4dcgi/web_command?year=2007&month=jan”

$0:=”http://www.domain.com/4daction/web_command”

$0:=”http://maps.google.com/maps?near=3031+Tisch+Way,+San+Jose,+CA+95128”

When $0 is a JavaScript

$0:=”JS:alert(‘Hello San Jose);”

$0:=”JS:myWin = window.open (‘http://www.4d.com, '4D', 'width=500,height=500'
);”

$0:=”JS:myDataWin = new DataWindow(‘Customer’, $(‘myDiv’));”

A Dynamic DDW relies on a method that the developer defines to process and
create an appropriate response. The response for a Dynamic DDW must be defined
as text: C_TEXT($0). For example:

` Method: DDW_4D_WebSite
C_TEXT($0)
$0:=”http://www.4d.com”



Here is an example that returns JavaScript:

` Method DDW_4D_WebSite_Popup
C_TEXT($0)
$0:=”JS:myWin = window.open(‘http://www.4d.com, ‘4D’, 'width=500,height=500');”

Note: The URL returned by a DDW is not redirected by 4D. It is returned to the front-end
as an XML attribute. The 4DAF front-end must handle it separately.

A Static DDW is strictly for a DDW method that returns a Blob. The returned Blob
can contain HTML code or static text. This may sound opposite to the term “Static”
because the returned value can be anything. However, the reason why we consider
it static is because the 4DAF front-end does not have to further process what is
returned. The returned value will simply be displayed in a window when the front-
end receives it. The 4DAF front-end will assume that the result is text or HTML that
the browser will process.

To create a Static DDW method, $0 must be defined as a Blob: C_BLOB($0).

` Method: DDW_Greetings
C_BLOB($0)
TEXT TO BLOB("<b>Hello San Jose</b>";$0;Text without length)

` Method: DDW_ShowImage
C_BLOB($0)
C_PICTURE($pictureVar)
GET PICTURE FROM LIBRARY(2055;$pictureVar)
PICTURE TO BLOB($pictureVar;$0;"JPEG")
DAX_Dev_MimeType("image/jpeg")

DDW in 4DAF Client vs. your own HTML page

In the 4D Ajax Framework Client, a DDW can be created and used at the portal
level, selection level or field level. A portal level DDW is displayed as a link in the
left sidebar of the main window. A selection level DDW is displayed as a button at
the top of a record selection window, such as a Grid or Calendar. A field level DDW
causes the field data to display as a link in a list of records and as a button next to
the field value in the input form (Editor).

Please note that all DDWs are handled automatically by the 4D Ajax Framework
Client. Depending on the type of the DDW that the developer created, the 4D Ajax
Framework Client will handle the execution and result based on the calling point.

When creating your own HTML page, the DDW will be handled automatically from
within the default 4D Ajax Framework layout: Grid, DataMatrix, Editor, etc. If the
API and 4D Ajax Framework URL is used independently from the default
objects/layouts, the developers may need to handle the responses themselves.



Creating and installing a DDW through Developer Hooks

Installing a DDW in the back-end database is done in the method named
DAX_DevHook_DDW_Install. This method is executed during the startup of the
database. This method is a public component method so the developer can add
DDW creation code as needed.

There are two methods that the developer will use to create and associate a DDW
to a particular area of 4D Ajax Framework: DAX_Dev_DDW_Create and
DAX_Dev_DDW_AssignToObject.

Syntax: DAX_Dev_DDW_Create                                                             
$0 LONG DDW ID
$1 TEXT Type of DDW (LinkStatic or LinkDynamic)
$2 TEXT Title of the DDW
$3 TEXT Method name or URL (absolute or relative to database Web root)
$4 TEXT Association Type (Portal or Other)

If $4 is "Portal" it cannot be assigned to a selection or field. 4D Ajax Framework
Client will immediately add it to the Portal level. The DDW will appear in the left
column of available links in the 4D Ajax Framework Client. If $4 is "Other" then the
DDW can be assigned to a selection or field.

Syntax: DAX_Dev_DDW_AssignToObject                                              
$1 LONG DDW ID
$2 TEXT Object (“Selection” or “Field”)
$3 TEXT Name of the Table
$4 TEXT Name of the Field (Required only when $2 = Field)

Create Portal DDW

Here is an example on how to create a DDW for the Portal.

` Method: DAX_DevHook_DDW_Install
C_LONGINT($ddwID_l)
$ddwID_l:=DAX_Dev_DDW_Create("LinkStatic";"4D";"http://www.4d.com";"Portal")
$ddwID_l:=DAX_Dev_DDW_Create("LinkStatic";"ShowImage";"DDW_ShowImage";"Portal")

Once the method DAX_DevHook_DDW_Install is executed at the startup the DDWs
are created and added to the Portal column of the 4D Ajax Framework Client.



Here is an example of the Portal with 2 DDWs installed:

Assign a DDW to a Selection or Field

The following is an example of how to associate a DDW to a Selection:

` Method: DAX_DevHook_DDW_Install
C_LONGINT($ddwID_l)
$ddwID_l:=DAX_Dev_DDW_Create("LinkStatic";"Summary";"GetSummary";"Other")
DAX_Dev_DDW_AssignToObject($ddwID_l;"Selection";"Invoices")

The DDW option will appear as a button at the top of the output layout of the
selection.

 

Assigning a DDW to a field is done in the same way as a Selection. Here is an
example:

` Method: DAX_DevHook_DDW_Install
C_LONGINT($ddwID_l)
$ddwID_l:=DAX_Dev_DDW_Create("LinkStatic";"Show Item";"GetItems";"Other")
DAX_Dev_DDW_AssignToObject($ddwID_l;"Field";"Invoices";"ID")

Once the DDW is created and assigned to [Invoices]ID, the DDW button will appear
in the input layout (Editor) next to the ID field for an existing record. It will not
appear when the input layout is opened for a new record, until the record is saved.



Here is an example:

The DDW also causes the value in [Invoices]ID to appear as a link in the output
layout. The link allows the user to execute the DDW without having to first open the
input layout.

Callbacks
------------------------------------------------------------------------------------------------------------------------------------------------------------------

Callbacks are similar to a form event in 4D. Callbacks allow certain actions to take
place when a particular event occurs. Normally you will use a callback to perform
some type of data manipulation in a field. In 4D Ajax Framework 1.x, the back-end
callback can be triggered by two events: On Load and On Data Change.

Note: The callback execution will happen at the field level only and it is supported from
within the input layout only.

In version 1.0 and 1.1, all installation of a callback is done inside the method
DAX_DevHook_InstallCallBack. When a callback is triggered, the method
associated with the field will be executed. However, its ability is limited to only
manipulating the data in its own field. We wanted to expand this feature to all
fields, so starting in version 1.2 a callback can be used to manipulate data in any
fields in the input layout. This means that when a callback is triggered on any single
field you can manipulate the current value of any of the other fields in the Editor.
This extended feature requires that we add additional DevHook methods for the
callback: DAX_DevHook_CB_Install and DAX_DevHook_CB_EventFired.

For version 1.2 and later, the previous Callbacks implementation has been
deprecated. Developers should use the newer method DAX_DevHook_CB_Install to
install all programmatic callbacks, and DAX_DevHook_CB_EventFired to handle all
callback events. Callbacks installed in an earlier version of 4D Ajax Framework will
continue to work in version 1.2, but they should be modified to use the newer
methods as soon as possible. Support for the older callback methods may be
removed in a later version of 4DAF.



DAX_DevHook_CB_Install is executed at database startup. Within this method, the
developer calls the method DAX_Dev_CB_Install to enable an event for a specific
field. Here are some installation examples:

 ` Method: DAX_DevHook_CB_Install
DAX_Dev_CB_Install(On Load;”Invoices”;”ID”) ` [Invoices]ID
DAX_Dev_CB_Install(On Load;”Customers”;”Name”) ` [Customers]Name
DAX_Dev_CB_Install(On Data Change;”Customers”;”Prefix”) ` [Customers]Prefix

To handle a callback, the developer must implement a trap inside the method
named DAX_DevHook_CB_EventFired. This method will be triggered for every
callback event. The developer is responsible for handling each event for each field
via their own custom code. The following is the information that is available to the
developer within this hook:

• Event that has fired
• Name of the selection that is being edited
• ID of the selection that is being edited
• Name of the field that is being edited
• ID of the field that is being edited
• Current value for the field that is being edited
• Record number that is being edited

This information can be obtained by calling the method DAX_Dev_CB_GetInfo.
Here is an example:

` Method: DAX_DevHook_CB_EventFired
$eventID_l:=Num(DAX_Dev_CB_GetInfo("event id"))
$selectionName_t:= DAX_Dev_CB_GetInfo("selection name")
$selectionID_l:=Num(DAX_Dev_CB_GetInfo("selection id"))
$fieldName_t:= DAX_Dev_CB_GetInfo("field name")
$fieldID_l:=Num(DAX_Dev_CB_GetInfo("field id"))
$fieldValue_t:= DAX_Dev_CB_GetInfo("field value")
$recordID_l:=Num(DAX_Dev_CB_GetInfo("record id"))

This information can be used to determine what would be the appropriate action for
the given callback. The following is example code based on the callbacks that we
installed in the method DAX_DevHook_CB_Install.

` Method: DAX_DevHook_CB_EventFired
Case of
  : ($eventID_l=On Load)
    Case of
      : ($selectionName_t=”Invoices”)
         If ($recordID_l=New record) & ($fieldName_t=”ID”)
           ` Perform Data Manipulation here
         End if
      : ($selectionName_t=”Customers”)
         If ($recordID_l#New record) & ($fieldName_t=”Name”)
           ` Perform Data Manipulation here
         End if
    End case
    DAX_Dev_CB_SetStatus(1)

  : ($eventID_l=On Data Change)



    Case of
       : ($selectionName_t=”Customers”)
         If ($fieldName_t=”Prefix”)
           ` Perform Data Manipulation here
         End if
    End case
    DAX_Dev_CB_SetStatus(1)

End case

The developer can manipulate the data in any of the fields in the input layout. This
is done by calling the method DAX_Dev_CB_SetFieldValues. This method takes
two parameters: a pointer to a Text array containing field names and a pointer to a
Text array containing field values. Here is an example based on the callbacks that
we installed in the method DAX_DevHook_CB_Install.

Case of
  : ($eventID_l=On Load)
    Case of
      : ($selectionName_t=”Invoices”)
         If ($recordID_l=New record) & ($fieldName_t=”ID”)
          ARRAY TEXT($fieldNames_at;1)

ARRAY TEXT($fieldValues_at;1)
$fieldNames_at{1}:=$fieldName_t
$fieldValues_at{1}:=Sequence number([Invoices])
DAX_Dev_CB_SetFieldValues(->$fieldNames_at;->$fieldValues_at)

         End if
      : ($selectionName_t=”Customers”)
         If ($recordID_l#New record) & ($fieldName_t=”Name”)
          ARRAY TEXT($fieldNames_at;1)

ARRAY TEXT($fieldValues_at;1)
$fieldNames_at{1}:=$fieldName_t
$fieldValues_at{1}:=Uppercase($fieldValue_t)
DAX_Dev_CB_SetFieldValues(->$fieldNames_at;->$fieldValues_at)

         End if
    End case
    DAX_Dev_CB_SetStatus(1)

  : ($eventID_l=On Data Change)
    Case of
       : ($selectionName_t=”Customers”)
         If ($fieldName_t=”Prefix”)
          ARRAY TEXT($fieldNames_at;1)

ARRAY TEXT($fieldValues_at;1)
$fieldNames_at{1}:=”Sex”

             If ($fieldValue_t=”Mr.”)
                $fieldValues_at{1}:=”Male”
             Else
                $fieldValues_at{1}:=”Female”
             End if

DAX_Dev_CB_SetFieldValues(->$fieldNames_at;->$fieldValues_at)
         End if
    End case
    DAX_Dev_CB_SetStatus(1)

End case

Callback status is set in the method DAX_Dev_CB_SetStatus. This method tells
the Web front-end whether the callback execution is successful or not. A failed



callback is indicated by changing the background color (Yellow) of the field that
triggered the callback.

DAX_Dev_CB_SetStatus(1) -> Success
DAX_Dev_CB_SetStatus(0) -> Fail

Along with the method DAX_Dev_CB_SetStatus, the developer can add a
meaningful message to the front-end by calling the method
DAX_Dev_CB_SetMessage. For example:

DAX_Dev_CB_SetStatus(0) ` Callback fail
DAX_Dev_CB_SetMessage("Invalid Zip Code") ` Give the reason why

Choice Lists
------------------------------------------------------------------------------------------------------------------------------------------------------------------

Using a choice list is a great way to prevent user errors in data entry. In 4D a
choice list can be set to a field directly at the structure level or it can be set to an
object in the input form. 4D Ajax Framework offers very similar capabilities. If a
choice list is set to a field at the structure level it will become automatically
available in the input layout of the 4D Ajax Framework. A choice list can also be
assigned to a field programmatically by using the Developer Hooks.

Programmatically set a choice list to a field

Setting a choice list to a field can be done programmatically from inside the method
named DAX_DevHook_InstallChoiceList. Within this method the developer must
call the method DAX_Dev_SetChoiceList to assign a list to a field. You can also
call the method DAX_Dev_SetChoiceList to override the current choice list with
another one if the field already has a list set to it at the structure level.

Syntax: DAX_Dev_SetChoiceList                                                       
$1 TEXT Table Name
$2 TEXT Field Name
$3 TEXT List Name (Must exist in the List Editor)
$4 BOOL True to guarantee that 4DAF use this list instead of the default

one

Examples

 ` Method: DAX_DevHook_InstallChoiceList
DAX_Dev_SetChoiceList(Table name(2);Field name(2;1);"States")

 ` Method: DAX_DevHook_InstallChoiceList
DAX_Dev_SetChoiceList("Location";"States";"States")

 ` Method: DAX_DevHook_InstallChoiceList
 ` Force 4DAF to use this list even if the
 ` field already has a choicelist assigned to it.
DAX_Dev_SetChoiceList(Table name(2);Field name(2;1);"States";True)



Note: The method DAX_DevHook_InstallChoiceList will be executed once at every launch
during database startup.

Programmatically alter the content of the list during callbacks

Programmatically altering the list contents can be very useful when the content of
the list depends on the current value of another field or some other condition that is
included in the request. There are two available places in 4D Ajax Framework that
the developers can programmatically modify the list: in DAX_DevHook_CB_
EventFired (during a callback) and DAX_DevHook_ListContents (before the list is
sent to the front-end).

During Callbacks

It might be easier to illustrate this idea through an example. Suppose you have an
input layout that allows the user to specify car make and model. Let us assume that
we have the name of all makes and models saved in the database. We can start by
setting a list with the name of all cars to the field named Make ([Car]Make). Since
the content of this list doesn’t have to change, we can set it from the method
DAX_DevHook_InstallChoiceList:

 ` Method: DAX_DevHook_InstallChoiceList
DAX_Dev_SetChoiceList("Cars";"Make";"Make_List")

Next we need to set an empty list to Model ([Cars]Model). Since the content will be
programmatically altered during callback, it does not have to contain anything at
this point. For this example, you can even specify a name to represent a list that
does not exist. The method DAX_DevHook_InstallChoiceList should now look like
this:

` Method: DAX_DevHook_InstallChoiceList
DAX_Dev_SetChoiceList("Cars";"Make";"Make_List")
DAX_Dev_SetChoiceList("Cars";"Model";"Model_List")

The next step is to change the list contents programmatically in the method
DAX_DevHook_CB_EventFired when the user causes the callback event to fire by
selecting a particular Make. The name of the method that the developer will call is
DAX_Dev_CB_SetChoiceList. This method does not change the actual content of
the list in the 4D List Editor. It tells 4D Ajax Framework to replace the named
choice list contents with the new content.

Syntax: DAX_Dev_CB_SetChoiceList                                                
$1 TEXT Name of the List
$2 POINTER Pointer to a Text array containing new list items

Here is an example of how DAX_DevHook_CB_EventFired would look like after we
added the dynamic list manipulation code into it.

` Method: DAX_DevHook_CB_EventFired
$eventID_l:=Num(DAX_Dev_CB_GetInfo("event id"))



$selectionName_t:=DAX_Dev_CB_GetInfo("selection name")
$fieldName_t:=DAX_Dev_CB_GetInfo("field name")
$fieldValue_t:=DAX_Dev_CB_GetInfo("field value")
$recordID_l:=Num(DAX_Dev_CB_GetInfo("record id"))
Case of

: ($eventID_l=On Load)
Case of

: ($selectionName_t=”Cars”)
If ($fieldValue_t#””)

If ($recordID_l=New record) & ($fieldName_t=”Make”)
ARRAY TEXT($listModel_at;0)
LoadCarModel($fieldValue_t;->$listModel_at)
DAX_Dev_CB_SetChoiceList("Model";->$listModel_at)

End if
End if

End case
DAX_Dev_CB_SetStatus(1)

: ($eventID_l=On Data Change)
Case of

: ($selectionName_t=”Cars”)
If ($fieldName_t=”Make”) & ($fieldValue_t#””)

ARRAY TEXT($listModel_at;0)
LoadCarModel($fieldValue_t;->$listModel_at)
DAX_Dev_CB_SetChoiceList("Model";->$listModel_at)

End if
End case
DAX_Dev_CB_SetStatus(1)

End case

Before list is sent to the front-end

Before a list is put into an XML response the developer can override the list items in
the Developer Hook DAX_DevHook_ListContents. Every time the 4D Ajax
Framework is about send a list to the front-end the developer is given the
opportunity to modify the list. This allows you to remove items from the list or
create the list completely in code.

Syntax: DAX_DevHook_ListContents                                                
$1 TEXT Name of the list that has been requested
$2 POINTER Pointer to a Text array containing the list items

Here is a basic example:

 ` Method: DAX_DevHook_ListContents
$list_t:=$1
$listItems_p:=$2
Case of
  : ($list_t="My_List")
    C_LONGINT($find_l)
    $find_l:=Find in array($listItems_p->;"Remove_Value")
    If ($find_l#-1)
      DELETE ELEMENT($listItems_p->;$find_l;1)
    End if
  : ($list_t="My_Other_List")
    READ ONLY([My_Table])
    ALL RECORDS([My_Table])



    DISTINCT VALUES([My_Table]My_Field;$listItems_p->)
    UNLOAD RECORD([My_Table])
End case

Here is another example using the Cars scenario from the last section:

` Method: DAX_DevHook_ListContents
$list_t:=$1
$listItems_p:=$2
Case of
  : ($list_t="Model_List")
     ` Remove all car models that are no longer in stock
    FilterOutOfStockModel($listItems_p)

  : ($list_t="Make_List")
    ` Sort all makes by Rank
    SortCarByRank($listItems_p)

End case

Summary
------------------------------------------------------------------------------------------------------------------------------------------------------------------

The 4D Ajax Framework Developer Hooks provide 4D developers the necessary
tools to allow them to take control of many of the standard behaviors. The
discussion in this Technical Note was primarily on the four main developer hooks:
Developer Created Selections, Developer Defined Windows, Callbacks and Choice
Lists. This Technical Note should clarify and provide many ideas on how you can
use this powerful feature.

A Note about 4D Web 2.0 Pack
------------------------------------------------------------------------------------------------------------------------------------------------------------------

The products in 4D Web 2.0 Pack are a departure from most other 4D products. As
4D Web 2.0 Pack is a subscription-based product it is expected that incremental
releases will be made. Thus please note that this Technical Note is based on 4D
Ajax Framework 1.2. As new features are implemented this Technical Note may
become obsolete (faster than most other 4D products). For the latest information
on the 4D Ajax Framework consult the latest documentation and also check the 4D
Web 2.0 Pack Wiki.

Related Resources
------------------------------------------------------------------------------------------------------------------------------------------------------------------

4D Web 2.0 Pack documentation downloads:
http://www.4d.com/support/documentation.html#w20p

4D Web 2.0 Pack Wiki:
http://daxipedia.4d.com

4DAF Demos:
http://www.4d.com/Products/4dweb20pack_demos.html


